

Министерство общего и профессионального образования  
Российской Федерации

МАТИ – Российский Государственный технологический  
Университет им. К. Э. Циолковского

---

Кафедра “ Высшая математика”

## **Обработка данных, хранящихся в текстовых файлах**

Методические указания к лабораторной работе  
по курсу “Информатика”

Составили **Сидоров Б. Н. Никулин А. М.**

Москва 1998

Данное руководство предназначено для студентов, изучающих язык Паскаль и выполняющих лабораторные работы по курсу "Информатика". Руководство преследует цель привить практические навыки в составлении программ на языке Паскаль и их выполнение на машине типа IBM PC .

## Файлы в Турбо-Паскале.

Файл – именованная область диска. Всё , что хранится на диске хранится в виде файлов.

Преимущество использования в программах:

- сохранения результатов работы программы
- при обработке большого объема данных, во избежании их повторного ввода.

Текстовый файл содержит любую последовательность символов, объединённую в строки произвольной длины.(слова, числа, знаки...)

Для работы с текстовым файлом необходимо:

.Описать переменную файлового типа в программе:

```
var f: text; { вместо f может быть любое разрешённое имя турбо-Паскаля)
```

.Связать описанную переменную с файлом на диске это делается процедурой:

```
assign(f,' имя файла '), где f – имя файловой переменной в программе,  
имя файла – имя файла на диске.
```

.Открыть файл

а) Для чтения.

Для чтения можно открыть только существующий файл; это делается процедурой

```
reset(f)
```

указатель файла устанавливается на первую запись. Доступ к файлу осуществляется последовательно от записи к записи, указатель указывает на текущую доступную запись.

б) Для записи.

Для записи открывается файл, который ещё не существует на диске , при этом он создается, если же файл с таким именем уже был, то его содержимое уничтожается. Открытие для записи производится процедурой

```
rewrite(f);
```

с) Для дозаписи ( без потери информации).

Открытие файла для дозаписи производится процедурой

```
append(f);
```

указатель файла устанавливается на последнюю запись.

Чтение из файла осуществляется оператором

readln (f, список ввода);

если перед списком ввода в операторе readln стоит файловая переменная, то он вводит данные не с клавиатуры, а из файла.

**Запись** в файл осуществляется оператором

writeln (f, список вывода); или write (f, список вывода);

в котором перед списком вывода стоит файловая переменная.

**Завершение** работы с файлом – закрытие файла осуществляется процедурой

*close(f);*

### **Дополнительные функции для работы с файлами.**

Функция *eof(f):boolean*; Логическая функция, возвращает значение true, если файловый указатель стоит в конце файла.:

Функция *eoln(f):boolean*; Логическая функция, возвращает значение true, если файловый указатель стоит в конце строки файла.

### **Конструкция**

while not eof(f) do readln(f, список переменных);

позволяет обрабатывать файл, пока в нём есть непросмотренные данные.

Функция *ioresult(f):integer*; Возвращает признак последней операции ввода-вывода. Если операция завершилась успешно, функция выдает ноль. Функция доступна только при отключенном контроле ошибок ввода-вывода. Директива компилятора {\$I-} отключает, а {\$I+} включает контроль ошибок ввода-вывода.

**Функция** используется для предотвращения аварийного завершения программы при попытке открытия для чтения несуществующего файла

{\$I-}

reset(f)

{\$I+}

if ioresult<>0 then (файл не существует ) else { файл существует }

## Примеры работы с файлами

**Пример1.** Создать текстовый файл, записав в него случайные числа из диапазона -20...20

```
uses crt;
const n=16;
var i,x:integer;
  f:text; {описание файловой переменной}
begin
  {связывания файловой переменной с именем файла}
  assign(f,'aa1.txt');
  {открытие файла для записи - файл создается в текущем каталоге}
  rewrite(f);
  randomize;
  for i:=1 to n do
    begin
      {получение случайного числа из диапазона -20...20}
      x:=random(41)-20;
      {запись числа в файл}
      writeln(f,x);
      {вывод числа на экран}
      writeln(x);
    end;
  close(f);
  repeat until keypressed;
end.
```

**Пример 2.** Найти сумму и количество положительных и отрицательных чисел из файла, созданного в примере 1.

```
uses crt;
var i,kotr,kpol,sotr,spol,x:integer;
  rez:integer;
  f:text; {описание файловой переменной}
begin
  clrscr;
  {связывания файловой переменной с именем файла}
  assign(f,'aa1.txt');
  {открытие файла для чтения, если такого файла нет,
  то выход из программы }
```

```

{$I-}
reset(f);
{$I+}
rez:=ioresult;
if rez<>0 then
    begin
        writeln(' error, press ENTER ');
        readln;
        exit;
    end;
sotr:=0;spol:=0;kotr:=0;kpol:=0;
while not eof(f) do { пока нет конца файла }
    begin
        { чтение из файла числа,
        число автоматически преобразуется в тип integer }
        readln(f,x);
        { вывод числа на экран }
        writeln(x);
        { если число отрицательное, то оно добавляется в переменную
        sotr, содержащую сумму отрицательных чисел и
        увеличивается на 1 переменная kotr, содержащая их
        количество }
        if(x<0 ) then begin sotr:=sotr+x; kotr:=kotr+1; end;
        { если число положительное, то оно добавляется в переменную
        spol, содержащую сумму положительных чисел и
        увеличивается на 1 переменная , kpol содержащая их
        количество }
        if(x>=0) then begin spol:=spol+x; kpol:=kpol+1; end;
    end;
    { закрыть файл }
close(f);
    { вывод результата на экран }
    writeln('sotr= ',sotr:4,' kotr= ',kotr:3);
    writeln('spol= ',spol:4,' kpol= ',kpol:3);
    { открытие файла для дозаписи }
append(f);
    { вывод результата в файл }
    writeln(f,'sotr= ',sotr:4,' kotr= ',kotr:3);
    writeln(f,' spol= ',spol:4,' kpol= ',kpol:3) ;
    { закрыть файл }

```

```
close(f);
repeat until keypressed;
end.
```

Пример 3.

{ Программа создает текстовый файл, содержащей две колонки;  
первая - 15 поз. - дата; вторая - температура. }

```
uses crt;
```

```
const n=5;
```

```
var i:integer;
```

```
  x: integer;
```

```
  dat:string[15];
```

```
  mes:string[10];
```

```
  den:string[5];
```

```
  j,p:integer;
```

```
  f:text;
```

```
begin
```

```
  { связывание файловой переменной f с именем файла b.txt }
```

```
  assign(f,'b.txt');
```

```
  { открытие файла для записи - файл создается в текущем каталоге }
```

```
  rewrite(f);
```

```
  { ввод названия месяца }
```

```
  write(' Введите месяц '); readln(mes);
```

```
  { запись в файл строки - заголовка }
```

```
  writeln(f,'Температура за ',mes);
```

```
  for i:=1 to n do
```

```
    begin
```

```
      { ввод дня }
```

```
      write(' Введите день '); readln(den);
```

```
      { формирование даты - соединение двух строк дня и месяца }
```

```
      dat:=den+' '+mes;
```

```
      { дозапись пробелов в конец строки:
```

```
      строка должна состоять из 15 символов }
```

```
      p:=length(dat); { определение длины строки }
```

```
  for j:=p+1 to 15 do dat:=dat+' ';
```

```
    { ввод температуры }
```

```
    write(' Введите температуру за ',dat,' '); readln(x);
```

```

    { запись в файл }
    writeln(f,dat,x);
    end;
    { закрытие файла }
    close(f);
    writeln('Press ENTER');
    repeat until keypressed;
end.

```

Пример 4.

{ Программа считывает из файла строку, содержащую две колонки; первая - 15 поз. - дата; вторая - температура; и находит величины: среднюю температуру, кол. дней с положительной и отрицательной температурой и количество дней с температурой превышающей среднюю. }

```

uses crt;
var i,k1,k2,s,x,kol:integer;
    rez,cod:integer;
    sred:real;
    st:string;
    dat:string[15];
    f:text;
begin
    clrscr;
    { связывание файловой переменной f с именем файла b1.txt }
    assign(f,'b.txt');
    { открытие файла для чтения, если такого файла нет,
      то выход из программы (процедура exit) }
    {$I-}
    reset(f);
    {$I+}
    rez:=ioresult;

```

```

if rez<>0 then
    begin
        writeln(' error, press ENTER ');
        readln;
    end;

```

```

        exit;
    end;
    { чтение из файла и вывод на экран первой строки - заголовка }
    readln(f,st); writeln(st);
    { обнуление переменных, используемых при решении задачи
      i - количество дней
      s - сумма температур всех дней
      k1 - количество дней с отрицательной температурой
      k2 - количество дней с положительной температурой }
    i:=0;s:=0;k1:=0;k2:=0;
    { обработка строк, пока нет конца файла }
    while not eof(f) do
        begin
            { читается строка из состоящая из
              dat - даты типа string[15] и x - температуры целого типа }
            readln(f,dat,x);
            { вывод прочитанной строки на экран }
            writeln(dat,' ',x);
            { обработка данных }
            i:=i+1;
            s:=s+x;
            if(x<0) then begin k1:=k1+1; end;
            if(x>=0) then begin k2:=k2+1; end;
        end;
    sred:=s/i; { средняя температура }
    { закрыть файл }
    close(f);
    { вывод результата }
    writeln(' кол. дней с отрицательной темпер. = ',k1:3);
    writeln(' кол. дней с положительной темпер. = ',k2:3);
    writeln(' средняя температура = ',sred:4:1);
    { открытие файла для повторного чтения }
    reset(f);
    readln(f,st);
    { вычисление количества дней с температурой,превышающей среднюю }
    }
    kol:=0;
    while not eof(f) do
        begin
            readln(f,dat,x);

```

```

    if(x>=sred) then begin kol:=kol+1; end;
end;
writeln('количество дней с температурой, превышающей среднюю
',kol);
writeln(' Press ENTER');
repeat until keypressed;
end.

```

{ **Пример 5..**

Исходная информация: файл, строки которого содержат следующие поля: фамилия и имя ученика (25 позиций), его пол (1 позиция м/ж) количество оценок и сами оценки.

Составить программу, определяющую

- 1) ученика с наибольшим средним балом.
  - 2) всех учеников, учащихся на 4 и 5 и пересылающих их в новый файл
- Исходный файл создается в текстовом редакторе NC

Доброва Ира	ж	4	4	3	2	3	
Дробная Юлия	ж	5	3	3	2	5	4
Звягин Алексей	м	3	4	3	5		
Ильюшенко Катя	ж	4	4	4	5	4	
Капцевич Ольга	ж	3	2	3	4		
Комогоров Николай	м	6	3	4	3	5	3
Мирошкин Александр	м	5	3	5	4	3	4
Скрипкин-Кузнецов Михаил	м	6	4	5	4	5	4
Фетисов Денис	м	6	2	4	5	3	3
Черноусова Лиза	ж	4	4	5	4	4	

```

uses crt;
var i,j,kol,r,sum,bal:integer;
    sr,max:real;
    rez,cod:integer;
    fam,xar,kom,fmax:string[25];
    pol:char;
    t:boolean;{ t - true,если ученик учится на 4 и 5 ; t - false, если есть 2
или 3 }
    f,fn:text;
begin
    clrscr;
    { связывание файловой переменной f с именем файла s.txt }
    assign(f,'d:\tp\klass.txt');
    { открытие файла для чтения, если такого файла нет,
    то выход из программы (процедура exit) }
    {$I+}
    reset(f);
    {$I+}
    rez:=ioresult;
    if rez<>0 then
        begin
            writeln(' error ',rez, ' press ENTER ');
            readln;
            exit;
        end;
    { связывание файловой переменной fn с именем файла sc.txt }
    assign(fn,'klassotl.txt');
    { открытие файла для записи }
    rewrite(fn);
    { первоначально максимальный средний бал равен нулю }
    max:=0;
    { обработка строк, пока нет конца файла }
    while not eof(f) do
        begin
            { начальные значения для ученика }
            sum:=0; t:=true; xar:="";
            { читается часть строка из состоящая из
            fam - фамилии и имени ученика типа string[25]

```

```

pol - пол
kol - количество отметок    }
read(f,fam,pol,kol);
{ чтение и обработка отметок ученика }
for i:=1 to kol do
  begin
    read(f,r);
    sum:=sum+r;
    { если 2 или 3 то }
    if((r=2)or(r=3))then t:=false;
  end;
  { переход к новой строке в файле - обязательно! }
  readln(f);
  { вычисление среднего бала и сравнение с максимальным }
  sr:=sum/kol;
  { отметка в четверти }
  bal:=round(sr);
  if sr>max then begin max:=sr;fmax:=fam;end;
  { если у ученика нет 2 и 3 (t=true),
  то его характеристика - хорошо и пересылаем
  ученика в новый файл, иначе его характеристика - плохо }
  if t then
    begin хар:=' хорошо '; writeln(fn,fam,'отметка: ',bal); end
    else хар:=' плохо ';
  { вывод на экран ученика ,ср бала, отметки и характеристики }
  writeln(fam,'sr=',sr:5:2,' итог - ',bal ,хар);
end;
{ закрыть файл }
close(f);close(fn);
{ вывод результата }
writeln(' лучший ученик ',fmax);
writeln(' Press ENTER');
repeat until keypressed;
end.
Созданный файл:

```

Ильюшенкова Катя	отметка: 4
Скрипкин-Кузнецов Михаил	отметка: 5
Черноусова Лиза	отметка: 4

{ Пример 6..

Исходная информация: файл, строки которого содержат следующие поля: фамилия и имя ученика (25 позиций), его пол (1 позиция м/ж) количество оценок и сами оценки.

Составить программу, сортирующих учеников в порядке возрастания их среднего бала

Исходный файл создается в текстовом редакторе NC

Доброва Ира	ж 4 4 3 2 3
Дробная Юлия	ж 5 3 3 2 5 4
Звягин Алексей	м 3 4 3 5
Ильюшенкова Катя	ж 4 4 4 5 4
Капцевич Ольга	ж 3 2 3 4
Комогоров Николай	м 6 3 4 3 5 3 5
Мирошкин Александр	м 5 3 5 4 3 4
Скрипкин-Кузнецов Михаил	м 6 4 5 4 5 4 5
Фетисов Денис	м 6 2 4 5 3 3 4
Черноусова Лиза	ж 4 4 5 4 4

Пример с использованием массива }

```
uses crt;
const n=10;m=10;
var i,j,sum,k,im:integer;
    min:real;
    fam:string[25];
    list1:array[1..n] of string[25];{массив фамилий }
    list2:array[1..n] of string[1];{ массив, содержащий пол ученика }
    list3:array[1..n] of integer;{ массив, содержащий количество оценок
ученика }
    list4:array[1..n,1..m] of integer;{ массив оценок }
    result:array[1..n] of real; { массив, содержащий средний бал ученика }
    f:text;
begin
    clrscr;
```

```
{----- Открытие файла ----- }
{ связывание файловой переменной f с именем файла klass.txt }
```

```

assign(f,'klass.txt');
{ открытие файла для чтения, если такого файла нет,
  то выход из программы (процедура exit) }
{$I-}
reset(f);
{$I+}
if ioreult<>0 then
  begin
    writeln(' Ошибка нет файла - press ENTER ');
    readln;
    exit;
  end;
{----- чтение из файла ----- }
i:=1;
while not EOF(f) and (i<=n) do
  begin
    { читается часть строка из состоящая из
      list1[i] - фамилия и имя ученика типа string[25]
      list2[i] - пол
      list3[i] - количество отметок    }
    read(f,list1[i],list2[i],list3[i]);
    { чтение отметок ученика }
    for j:=1 to list3[i] do read (f,list4[i,j]);
    { переход к новой строке в файле( следующий ученик) -
      обязательно! }
    readln(f);
    i:=i+1;
  end;
close(f);
{ k - количество учеников в файле }
k:=i-1;
{ печать созданных массивов }
for i:=1 to k do
  begin
    write(list1[i],list2[i],' ');
    for j:=1 to list3[i] do write(list4[i,j],' ');
    writeln;
  end;
{ ----- обработка ----- }
{--1. вычисление среднего бала    формируется массив result}

```

```

for i:=1 to k do
  begin
    sum:=0;
    for j:=1 to list3[i] do
      sum:=sum+list4[i,j];
    result[i]:=sum/list3[i];
  end;
{ --2. сортировка массива result с перестановкой в массиве list1 }
for i:=1 to k-1 do
  begin
    { поиск минимума в i... k части массива }
    min:=result[i];im:=i;
    for j:=i+1 to k do
      begin
        if (result[j]<min ) then
          begin
            min:=result[j];
            im:=j;
          end;
      end;
    { перестановка }
    result[im]:=result[i]; result[i]:=min;
    fam:=list1[i]; list1[i]:=list1[im];list1[im]:=fam;
  end;
{ ----- ВЫВОД РЕЗУЛЬТАТА----- }
for i:=1 to k do writeln(list1[i],' итог - ',round(result[i]) );
writeln(' Press ENTER');
repeat until keypressed;
end.

```

#### Пример 7.

Исходная информация: файл, строки которого содержат следующие поля: фамилия и имя ученика , его пол ( м/ж) и оценка, полученная им на контрольной работе.

Составить программу, выводящую список всех учениц, имеющих оценку выше, чем средняя по классу

Исходный файл создается в текстовом редакторе NC

```

Доброва Ира ----- ж    4
Дробная Юлия ----- ж    2

```

Звягин Алексей м 2  
 Ильюшенкова Катя ж 4  
 Капцевич Ольга ж 2  
 Комогоров Николай м 4  
 Мирошкин Александр м 3  
 Скрипкин-Кузнецов Михаил м 5  
 Фетисов Денис м 4  
 Черноусова Лиза ж 4 }

```

uses crt;
const n=10;
var i,j,k,kol,kod,dlin:integer;
    result:real;
    s,a:string;
    list:array[1..n] of string; {массив, содержащий строки с учениками }
    list1:array[1..n] of string[25]; {массив фамилий }
    list2:array[1..n] of string; { массив имен }
    list3:array[1..n] of string; { массив, содержащий полученика }
    list4:array[1..n] of integer; {массив оценок }
    f:text;
begin
    clrscr;
    {----- Открытие файла ----- }
    { связывание файловой переменной f с именем файла klassa.txt }
    assign(f,'klassa.txt');
    { открытие файла для чтения }
    reset(f);

    {----- чтение из файла ----- }
    kol:=0;
    while not EOF(f) and (i<=n) do
      begin
        { читается вся строка целиком }
        kol:=kol+1;
        readln(f,list[kol]);
      end;
    close(f);
    { kol - количество учеников в файле }
  
```

```

{ ----- обработка ----- }
  for j:=1 to kol do
    begin
      s:=list[j];
      dlin:=length(s); { длина строки }
      i:=1; { i - указывает на считываемый символ в строке s }
      k:=0; { k - номер считываемого слова }
      {-- обработка строки пока номер считываемого символа меньше длины
строки }
      while i<=dlin do
        begin
          a="";
          { пропуск пробелов между словами }
          while(s[i]=' ') and (i<=dlin) do i:=i+1;
          k:=k+1; { увеличиваем счетчик числа слов }
          { пока не встретился пробел и не достигнут конец строки }
          while (s[i]<>' ') and(i<=dlin) do
            begin
              a:=a+s[i]; { символ s[i] добавляется в строку a }
              i:=i+1; { переход к следующему символу }
            end;
            { слово считано – помещаем его в массив }
            if k=1 then list1[j]:=a;
            if k=2 then list2[j]:=a;
            if k=3 then list3[j]:=a;
            if k=4 then begin val(a,list4[j],kod) end;
          end; { конец обработки строки }
        end; { конец обработки учеников }

{----- вычисление среднего бала ----- }
for i:=1 to kol do result:=result+list4[i];
result:=result/kol;
{ ----- вывод результата----- }
writeln(' средний бал класса -- ',result:4:1);
writeln(' список лучших учениц: ');
for i:=1 to kol do
  if (list4[i]>result) and (list3[i]='ж') then
    writeln(list1[i],',',list2[i],', --- ',list4[i] );
writeln(' Press ENTER');
repeat until keypressed; end.

```

## Задачи для самостоятельного решения

Дан Файл со структурой:

Первое поле – текстовое : фамилия автора, занимающее 10 позицией;

Второе поле – текстовое : название книги, занимающее 20 позицией;

Третье пол – текстовое : тематика книги ( худ./научн./комп. ),  
занимающее 5 позицией;

Четвертое поле – числовое : количество книг;

Пятое поле – числовое : стоимость одной книги;

Шестое поле – числовое : год издания (1990 –1998);

Необходимо составить программу

1. Определяющую книгу, наибольшей суммарной стоимости(стоимость\*количество)
2. Определяющую автора, имеющего наибольшее количество экземпляров книг.
3. Определяющую автора, имеющего наибольшее количество наименований книг
4. Определяющую количество наименований книг по каждой тематики
5. Определяющую тематику количество экземпляров ниг по которой больше
6. Определяющую книгу наибольшей стоимости
7. Определяющую книги количество экземпляров которых больше
8. Определяющую тематику книги по которой самые дорогие
9. Определяющую год за который представлено наибольшее количество книг.
10. Определяющую год, книги изданные в который самые дорогие
11. Определяющую год суммарная стоимость книг (стоимость\*количество) за которой наибольшая
12. Выводящую список всех книг, встречающихся в единственном экземпляре.
13. Выводящую список всех книг, встречающихся более чем в двух экземплярах.
14. Выводящую список всех авторов (автор должен быть представлен в списке только один раз) с указанием общего количества наименований его книг.
15. Выводящую список всех авторов (автор должен быть представлен в списке только один раз) с указанием общего количества экземпляров его книг.

16. Выводящую список авторов, имеющих книги более, чем по одной тематике
17. Выводящую список авторов, издававшихся только в последние три года.
18. Выводящую список книг, изданных по данной тематике в данный год ( тематика и год вводятся с клавиатуры).

## Литература.

1. В. Попов Turbo Pascal 7.0 для школьников М.,” Финансы и статистика”, 1996. -464 стр
2. Фаронов В.В. Основы Турбо-Паскаля М.: Изд-во МВТУ-ФЕСТО ДИДАКТИК , 1992.